Structural Nonlinearities

Exercise Supplement

for Release 5.5

001156 (Second Edition)
Volume II
April 30, 1999



ANSYS

Structural Nonlinearities for Release 5.5

Exercise Supplement

Table of Contents

Nonlinear Solution

Planar Seal Exercise

Exercise

To Do: **To become familiar with basic solution controls, the output file, and the monitor file, analyze the hyperelastic planar seal for an imposed deflection of 0.85" on the top rigid surface.**



This model has geometric (large deformation and large strain), material (hyperelasticity), and changing status (contact) nonlinear behavior.



0.42" Deflection

0.85" Deflection

Exercise

Steps to Follow:

1. Resume the database seal.db (the database includes a half symmetry model of the seal). The mesh and fixed boundary conditions for this analysis are included.
2. Enter solution and impose 0.85" of deflection in the -Y direction at the pilot node. (There is a parameter *n_load* in the database to identify the node.)
3. Activate nonlinear geometry (NLGEOM).
4. Set "time" equal to the absolute value of the imposed displacement.
5. Set the number of substeps for automatic time stepping. Try NSUBST,25,200,10. (If you have time you can experiment with the number of substeps to see how this affects the convergence behavior of the problem.)

Exercise

Steps to Follow:

6. Monitor the reaction force at the pilot node (*n_load*) as variable 3 in the monitor file.
7. Set the output controls to write every substep to the .rst file (OUTRES).
8. Solve the problem.
9. Review the results in the general postprocessor. Plot the deformed shape, stress results, and contact pressures.
10. Plot the force deflection curve in the time history postprocessor.

Exercise

Steps to Follow:

11. Review the output file and the monitor file. At what point in the solution does the program have difficulty converging? Why?

Extra Credit

If you have time rerun the analysis including friction. (Set MU = 0.2 for material 1.) Note the convergence behavior. Does including friction take more or less iterations? Plot the force deflection curve. What happens to the reaction force required to compress the seal?

Exercise

Answers to Questions:

At about 0.7” of deflection the program bisects to a smaller load increment to resolve the contact at the edge of the seal.

Contact initiates at the edge of the seal. Also note that at this point in the solution the reaction force increases.



Exercise

Answers to Questions:

Including friction requires more iterations. The solution becomes path dependent and another nonlinear effect is added to the problem.

Including friction increases the reaction force required to compress the seal.

Solution to Exercise

```
/BATCH
/PREP7
!*
ET,1,56
ET,2,169
ET,3,171
!*
MP,NUXY,1,0.49
TB,MOONEY,1,2
TBDATA,1,80
TBDATA,2,20
MP,MU,1,0.2
!*
R,1,,,0.5,0.1
R,2,,,0.5,0.1
!*
K,1
K,2,0.333,0
K,3,0.867,0.867
K,4,1.1,0.867
K,5,1.1,1
K,6,0.8,1
K,7,0.267,0.133
K,8,0,0.133
L,1,2
*REPEAT,7,1,1
L,8,1
LFIL,1,2,0.20
LFIL,2,3,0.15
LFIL,5,6,0.20
LFIL,6,7,0.15
LFIL,7,8,0.05
LFIL,8,1,0.05
AL,ALL
!*
K,25,-0.6,0
K,26,1.1,0
LSTR,26,25
K,27,-0.6,1.0
K,28,1.1,1.0
LSTR,27,28
!*
LESIZE,8,,,2
LESIZE,13,,,4
LESIZE,14,,,4
ESIZE,0.035
TYPE,1
MAT,1
AMESH,ALL
!*
LESIZE,15,,,1
LESIZE,16,,,1
TYPE,2
REAL,1
MAT,1
LMESH,15
LSEL,S,LINE,,1,3,1
LSEL,A,LINE,,9,10,1
LSEL,A,LINE,,14,14,1
NSLL,S,1
TYPE,3
ESURF
LSEL,ALL
TYPE,2
REAL,2
LMESH,16
KMESH,27
LSEL,S,LINE,,5,7,1
LSEL,A,LINE,,11,13,1
NSLL,S,1
TYPE,3
ESURF
LSEL,ALL
NSEL,ALL
DL,4,,UX,0
N_LOAD=NODE(KX(27),KY(27),0)
FINI
/SOLU
NLGEOM,ON
SOLC,ON
TIME,0.85
D,N_LOAD,UY,-0.85
NSUBST,25,200,10
OUTRES,ALL,ALL
MONITOR,VAR3,N_LOAD,FY
SOLVE
FINI
/EXIT
```

Geometric Nonlinearities

Upset Forging Exercise

Exercise

To Do: To become familiar with large strain solutions, analyze the upset forging of an axisymmetric billet.



- **This problem has geometric (large deformation and large strain), material (plasticity), and changing status (contact) nonlinear behavior.**
- **Accumulated equivalent plastic strains in the formed billet.**

ANSYS

Exercise

Steps to Follow:

1. Resume the database upset.db. The finite element mesh and fixed boundary conditions for this analysis are included in the database.

2. Enter solution and impose -9.0 mm of deflection in the UY direction at the pilot node. (There is a parameter n_load in the database to identify the pilot node.)

3. Activate nonlinear geometry (NLGEOM,ON).

4. Set "time" equal to the absolute value of the imposed displacement.

5 Set the number of substeps for automatic time stepping. Try NSUBST,100,1000,20.

Exercise

Steps to Follow:

6. Monitor the FY reaction force at the pilot node (n_load) as variable 1 in the monitor file.

7. Set the output controls to write ten evenly spaced solutions to the .rst file (OUTRES). (Every N^{th} substep, N = -10)

8. Solve the problem.

9. Review the results in the general postprocessor. Plot the deformed shape, and stress results. Review the level of element distortion (PlotCtrls > Style > Edge Options). Can you identify any potential problems with the level of element distortion?

Exercise

Steps to Follow:

10. Plot the force deflection curve in the time history postprocessor.

11. Review the output file and the monitor file. Note that for certain substeps there is a warning that the plastic strain limit is exceeded. How could the number of substeps be modified to reduce the plastic strain increment?

Exercise

Extra Credit

- **If you have time restart the analysis with an imposed displacement equal to 12.0 mm. Plot the force deflection curve. Why does the reaction force increase significantly with displacement?**



Exercise

Answers to Questions:

At 9.0 mm of deflection the elements at the "corner" of the billet are becoming distorted. The interior angles are approaching 180 degrees.

A secondary concern is the aspect ratio of the elements near the centerline.



Exercise

Answers to Questions:

- The minimum number of substeps could be increased to enforce a smaller load increment per substep (for example NSUBST,100,1000,40).
- Since plasticity is included in the problem the material behavior is nearly incompressible. The incompressibility caused by the large plastic strains result in the increased force required to compress the billet.

Solution to Exercise

```
/BATCH
RESUME,UPSET,DB
/SOLU
NLGEOM,ON
SOLC,ON
TIME,9
NSUBST,100,1000,20
OUTRES,ALL,-10
MONITOR,VAR1,N_LOAD,FY
D,N_LOAD,UY,-9
SOLVE
TIME,12
D,N_LOAD,UY,-12
SOLVE
FINISH
```

NOTES

Structural Stability

Cantilever Lateral Torsional Buckling Exercise

Exercise

To Do: Perform a pre-bucking analysis of a slender cantilever beam using eigenvalue and nonlinear buckling.



Material Properties:

EX = 10E6 psi

Poisson's ratio = 0.33

Loading:

P = 1 lb.

Classical Solution

P_{cr} = 31.856 lb.

Exercise

Steps to Follow:

1. Resume the database buckle.db. The database includes the mesh and fixed boundary conditions.
2. Plot the finite element mesh and the boundary conditions. (Turn on display of element shapes based on real constants.)
3. Enter Solution and apply the 1 lb. load to the free end (keypoint 2) in the +Y direction.
4. Run a static analysis with the pre-stress flag set (PSTRES,ON) to create the stress stiffness matrix.
5. Exit and re-enter solution. Following the procedure outlined in the notes, run an eigenvalue bucking analysis extracting and expanding the first four buckling modes. *Be sure to request stress results.*

Exercise

Steps to Follow:

6. Review the buckled mode shapes and the load factors in the general postprocessor.
7. Include an initial imperfection for the nonlinear buckling analysis. Follow the steps in the notes to include an initial imperfection of 0.005 based on the first buckled mode shape (2% of the beam width).
8. *Save the database with the updated geometry.* (We will need the perturbed geometry for the arc-length exercise.)
9. Run a nonlinear buckling analysis of the structure. Reapply the load to the free end, using a value slightly higher than the eigenvalue buckling load (35 lb.)

Exercise

Steps to Follow:

10. Re-enter solution and specify the analysis type as static.
11. Turn on nonlinear geometry (NLGEOM,ON). Set the number of substeps (25,500,25), and request that every substep be written to the results file (OUTRES).
12. Turn on the line search (LNSRCH). Why?
13. Execute the nonlinear buckling solution.
14. Review the output file. Did the line search help the problem to converge?
15. Review the unconverged results and the last converged solution in the general postprocessor.

Exercise

Steps to Follow:

16. Plot the load deflection curve in the time history postprocessor. Be sure to plot the applied load versus the lateral deflection.



Exercise

Answers to Questions:

- If you try running the problem without using the line search you will notice that the `MAX DOF INC` is oscillating. Activating the line search for this behavior helps to speed convergence.

- Activating the line search helps the problem to converge before the limit load is reached. The line search parameter is less than one for many of the equilibrium iterations.

Solution to Exercise

```
/BATCH
/PREP7
K,1,0,0,0,
K,2,100.0,0,0,
K,3,50,5,0,
LSTR,1,2
ET,1,BEAM189
SECTYPE,1, BEAM, RECT,
SECDATA,0.25,5.0
SLIST,1,1
MP,EX,1,10E6
MP,NUXY,1,0.33
LSEL,S,,,1,1,1
LATT,1,,1,0,3,,1
LESIZE,all,,,10
SECNUM,1
LMESH,all
DK,1,,,,0,ALL,,
FK,2,FY,1.0
FINISH
/SOLU
PSTRES,ON
EQSLV,SPARSE
SOLVE
FINISH
/SOLU
ANTYPE,BUCKLE
BUCOPT,LANB,4
MXPAND,4,,,YES
SOLVE
FINISH
/PREP7
UPGEOM,0.005,1,1,file,rst
/SOLU
ANTYPE,STATIC
OUTRES,ALL,ALL
NLGEOM,ON
LNSRCH,ON
NSUBST,25,500,25
FK,2,FY,35
SOLVE
FINISH
/EXIT
```

Structural Stability

Arc-Length Method Exercise

Exercise

To Do: Use the arc-length to determine the limit load for a slender cantilever beam.



B = 0.25
H = 5
L = 100

Material Properties:

EX = 10E6 psi

Poisson's ratio = 0.33

Loading:

P = 35 lb.

Classical Solution

P_{cr} = 31.856 lb.

Exercise

Steps to Follow:

1. Resume your database from the previous exercise with the geometric imperfection.
2. Apply 35 lb. of force at the free end (keypoint 2) in the +Y direction.
3. Enter solution and specify a static analysis.
4. Turn on nonlinear geometry (NLGEOM,ON).
5. Turn on the arc-length, let MINARC and MAXARC default.
6. Set a termination criteria. Set a displacement limit of 1.0" in the Z-direction at the free end (node 2).

Exercise

<u>Steps to Follow:</u>

7. Specify the number of substeps (NSUBST) to be 1,000.
8. Solve the problem.
9. Review the load deflection curve in the time history postprocessor.

Solution to Exercise

```
/BATCH
/PREP7
K,1,0,0,0,
K,2,100.0,0,0,
K,3,50,5,0,
LSTR,1,2
ET,1,BEAM189
SECTYPE,1, BEAM, RECT,
SECDATA,0.25,5.0
SLIST,1,1
MP,EX,1,10E6
MP,NUXY,1,0.33
LSEL,S,,,1,1,1
LATT,1,,1,0,3,,1
LESIZE,all,,,10
SECNUM,1
LMESH,all
DK,1,,,,0,ALL,,
FK,2,FY,35.0
FINISH
/SOLU
PSTRES,ON
EQSLV,SPARSE
SOLVE
FINISH
/SOLU
ANTYPE,BUCKLE
BUCOPT,LANB,4
MXPAND,4,,,YES
SOLVE
FINISH
/PREP7
UPGEOM,0.005,1,1,file,rst
/SOLU
ANTYPE,STATIC
OUTRES,ALL,ALL
NLGEOM,ON
ARCLEN,ON
ARCTRM,U,1.0,2,UZ
NSUBST,1000
SOLVE
FINISH
/EXIT
```

NOTES

Plasticity

Connector Exercise

Exercise

To Do: Analyze the socket connector for a proportional displacement controlled loading. Remove the imposed displacement and review the residual stresses.



A 2D plane stress with thickness model will be used to model the socket connector.



Exercise

Material Properties: Copper

EX = 16E6 psi

Poisson's ratio = 0.33



Strain	Stress
0.000625	10,000
0.0025	15,000
0.005	21,000
0.010	29,000
0.015	32,600
0.020	34,700
0.040	36,250
0.100	39,000

The data is presented as true stress-strain.

Exercise

Steps to Follow:

1. Resume the database connector.db, the database includes the mesh and fixed boundary conditions.
2. Review the element type selected for this analysis. Why is Plane42 selected for this analysis? Are there any other choices which are appropriate?
3. Enter the plasticity data table, Young's modulus and Poisson's ratio. What material model do you think is appropriate for this analysis?
4. Impose the 0.025" displacement in the -Y direction. (There is a parameter in the database n_load to identify the node with the imposed displacement.)

Exercise

Steps to Follow:

5. Turn on nonlinear geometry (NLGEOM,ON).
6. Specify the number of substeps (20,100,15).
7. Specify the output controls (request 10 evenly spaced solutions).
8. Solve
9. Review the accumulated equivalent plastic strain and stress results at 0.025" deflection in the general postprocessor.
10. Plot the load deflection curve in the time history postprocessor.

Exercise

Steps to Follow:

11. What is the reaction force at the imposed displacement?
12. Remove the imposed displacement and restart the solution.
    - In order to remove the imposed displacement it is necessary to switch from displacement control to force control. If imposed displacement is deleted (step changed) the model will instantaneously "spring back" due to the elastic energy and the solution will diverge. (You can try it.) To switch from displacement control to load control follow this procedure:
        - A. Obtain the reaction force at the imposed displacement. (Step 10)
        - B. Reenter solution and specify restart. Delete the imposed displacement and apply the reaction force to the same node. Turn off automatic time stepping and specify one substep. This is a null load step, and the solution should converge in one or two iterations.
        - C. Ramp the imposed reaction force to zero in load step 3. Turn on automatic time stepping, and specify the number of substeps (10,100,5), and the output controls.

Exercise

Steps to Follow:

13. Review the permanent deformation and residual stresses in the general postprocessor.



Permanent Deformation

Exercise

Answers to Questions:

Since this is a small strain plasticity analysis dominated by a bending response, Plane42 (full integration with incompatible modes) is selected.

Plane82 or Plane182 with reduced integration (a finer mesh would be required with Plane182) could also be selected. But, since we have applied a point constraint, hourglass behavior may be a problem with Plane182.

Since this a small strain analysis with eight data points multilinear kinematic hardening (KINH) will be used. (Although, because the load is proportional, multilinear isotropic hardening could also be used.)

Solution to Exercise

```
/BATCH
RESUME,connector,db
/PREP7
MP,EX,1,16e6
MP,NUXY,1,0.33,
TB,KINH,1, ,8, ,
TBPT,,0.000625,10000
TBPT,,0.0025,15000
TBPT,,0.005,21000
TBPT,,0.010,29000
TBPT,,0.015,32600
TBPT,,0.020,34700
TBPT,,0.040,36250
TBPT,,0.100,39000
/SOLUTION
ANTYPE,STATIC
NLGEOM,ON
SOLC,ON
TIME,1
NSUBST,20,100,15
OUTRES,ALL,-10
D,N_LOAD,UY,-0.025
SOLVE
FINI
/POST26
RFORCE,2,N_LOAD,F,Y,FY
PRVAR,2
*GET,NL_FY,VARI,2,EXTREM,VLAS
FINI
/SOLU
ANTYPE,,REST
TIME,2
DDEL,N_LOAD,UY
F,N_LOAD,FY,NL_FY
AUTOTS,OFF
NSUBST,1,1,1
SOLVE
TIME,3
F,N_LOAD,FY,0
NSUBST,10,100,5
SOLVE
FINI
/EXIT
```

NOTES

Hyperelasticity

O-Ring Exercise

Exercise

To Do: Analyze the rubber o-ring for an imposed displacement equal to half the diameter.



Finite element model of the hyperelastic o-ring.



Hydrostatic pressure with an imposed displacement of 0.060".

Exercise

Steps to Follow:

1. Resume the database oring.db, the database includes the finite element mesh and fixed boundary conditions.
2. Following the procedure outlined in the notes generate the Mooney-Rivlin constants for the material data provided (uniaxial tension/compression and planar tension). Use a two-term Mooney-Rivlin model. Why is a two term Mooney-Rivlin model selected?
3. Enter Poisson's ratio (0.49) for material 1.
4. Impose the 0.06" displacement in the -Y direction. (There is a parameter in the database n_load to identify the pilot node with the imposed displacement.)

Exercise

Uniaxial Tension/Compression

Strain	Stress
-0.6	-1521.0
-0.5	-840.0
-0.4	-493.6
-0.3	-291.1
-0.2	-160.1
-0.1	-68.4
0.0	0.0
0.1	53.7
0.2	97.7
0.3	135.1
0.4	167.8
0.5	197.0



Poisson's ratio

$\nu = 0.49$

Exercise

Planar Tension

Strain	Stress
0.0	0.0
0.2	124.3
0.4	207.1
0.6	271.2
0.8	325.7
1.0	375.0
1.2	421.2
1.4	465.4
1.6	508.6
1.8	550.9
2.0	592.6



Exercise

Steps to Follow:

5. Turn on nonlinear geometry (NLGEOM,ON).
6. Specify the number of substeps (10,100,5).
7. Specify the output controls (request every substep be written to the .rst file).
8. Review the stress and strain results at 0.06" deflection in the general postprocessor.
9. Plot the hydrostatic pressure using an element solution (PLESOL). Verify that the distribution is smooth.
10. Plot the load deflection curve in the time history postprocessor.

Exercise

Extra Credit

If you have time, restart the analysis and increase the imposed deflection to 0.085". Review the stress and strain distribution at 0.085" of deflection, also review the mesh distortion. Are your calculated material constants still valid for these levels of strain? What can you do to verify the material behavior?

Exercise

Answers to Questions

The engineering stress-strain data provided contains single curvature, therefore use a two-term Mooney-Rivlin model.

At 0.085" of deflection the strains are outside of the range of the material data provided. To show how the material behaves outside of the experimental range, the calculated stress-strain data can be plotted outside the experimental range. However, the only way to be sure the material behavior is correct is to obtain more test data!

Solution to Exercise

```
/BATCH
/PREP7
!
!  RID = INSIDE OD
!  RSD = SECTION OD
!
RID = 1.35
RSD = 0.12
!
ET,1,56,,,1
ET,2,169
ET,3,171
TB,HYPER,1
TBDATA,1,80.0
TBDATA,2,20.0
MP,NUXY,1,.49
MP,MU,2,0.0
R,2,,,1,0.1
R,3,,,1,0.1
LOCAL,11,1,RID+RSD/2
WPCSYS,11
PCIRC,RSD/2,,0,90
PCIRC,RSD/2,,90,180
PCIRC,RSD/2,,180,270
PCIRC,RSD/2,,270,360
NUMMRG,KP
MSHKEY,1
ESIZE,.015
AMESH,ALL
CSYS,0
K,20,RID-RSD,-RSD/2
K,21,RID+(2*RSD),-RSD/2
K,22,RID+(2*RSD),RSD/2
K,23,RID+(2*RSD),RSD/2
K,24,RID-RSD,RSD/2
L,21,20
L,22,21
L,24,23
TYPE,2
MAT,2
REAL,2
LMESH,6
LMESH,9
REAL,3
LMESH,11
KMESH,24
CSYS,11
NSEL,S,LOC,X,RSD/2
TYPE,3
REAL,2
ESURF
REAL,3
ESURF
ESEL,ALL
NSEL,ALL
D,ALL,UZ,0
N_LOAD=NODE(KX(24),KY(24
),0)
SAVE,ORING,DB
FINI
/SOLU
NLGEOM,ON
SOLC,ON
NSUBST,10,100,5
MONITOR,VAR3,N_LOAD,FY
OUTRES,ALL,ALL
DK,24,UY,-0.060
SOLVE
DK,24,UY,-0.085
SOLVE
FINI
/EXIT
```

ANSYS

Contact Nonlinearities

Snap Fit Exercise

Exercise

To Do: Analyze an axisymmetric snap fit assembly for a displacement controlled loading.



Snap Fit Assembly



Axisymmetric Model

Exercise

Steps to Follow:

1. Resume the database snap.db. The database includes the mesh and fixed boundary conditions.
2. Using the Contact Wizard create the target and contact elements as shown. Is it clear which side should be the target? Set the penalty stiffness scale factor (FKN) appropriately. Hint: Is the problem dominated by bending or bulk deformation? (We will neglect friction; do not enter a value for the coefficient of friction, MU.)



Exercise

Steps to Follow:

3. Impose the 0.4" displacement in the -Y direction on line 45.
4. Turn on nonlinear geometry (NLGEOM,ON).
5. Set "time" equal to 0.4 and specify the number of substeps for automatic time stepping (20,500,10).
6. Specify the output controls (request every substep).
7. Run the solution, and review the output and the monitor files.

Exercise

Steps to Follow:

8. Postprocess the contact results, review contact pressures and stress results. Animate the stress results.
9. Plot the load deflection curve in the time history postprocessor. Why do negative reaction forces develop in this problem?



Exercise

Extra Credit

- If you have time, restart the analysis and increase the deflection to 0.55" in the -Y direction to engage the second tooth of the snap fit assembly. Review the results and the convergence behavior of the problem.
- If you have time, rerun the analysis reversing the target and contact surface designations. Do the answers change significantly?



Exercise

Answers to Questions:

- In this problem it is not clear which sides should be the target and contact surfaces. In this case *symmetric* contact (both sides designated as target and contact surfaces) may be required. For this problem the side with the coarser mesh is selected as the target surface. (Although, one could argue that the surface designated as the contact surface is the simpler surface, and thus should be the target surface.)

- This problem is a bending dominated problem, set FKN = 0.1.

- Negative reaction forces develop because this solution is displacement controlled. At various points in the load history the imposed displacement causes negative reaction forces to develop.

Exercise

Answers to Questions:

For the first load step (0 to 0.4) the results do not vary significantly by switching the target and contact surface designations. For the second load step (0.4 to 0.55) the force deflection curve varies between the two cases. This is an indication that symmetric contact is required for this problem.

Solution to Exercise

```
/BATCH
RESUME,snap,db
/PREP7
ET,2,169
ET,3,171
R,1,,,0.1,0.1
CMSEL,S,TARGET
NSLL,S,1
TYPE,2
MAT,1
REAL,1
ESURF
CMSEL,S,CONTACT
NSLL,S,1
TYPE,3
ESURF
ALLS
FINISH
```

```
/SOLUTION
ANTYPE,STATIC
NLGEOM,ON
SOLC,ON
TIME,0.4
NSUBST,20,500,10
OUTRES,ALL,ALL
DL,45,,UY,-0.4
SOLVE
DL,45,,UY,-0.55
SOLVE
FINI
/EXIT
```

NOTES

Contact Nonlinearities

Hertz Contact Exercise

Exercise

To Do: Analyze a 2-D plane strain Hertz contact problem with large strain plasticity using a force controlled loading.



Exercise

Material Properties: Copper

EX = 16E6 psi

Poisson's ratio = 0.33



Strain	Stress
0.000625	10,000
0.0025	15,000
0.005	21,000
0.010	29,000
0.015	32,600
0.020	34,700
0.040	36,250
0.100	39,000
0.200	40,250

The data is presented as true stress-strain.

Exercise

Steps to Follow:

1. Resume the database hertz.db. The database includes the deformable finite element mesh and fixed boundary conditions.

2. Review the element type selected for the deformable mesh. Why is Plane182 with reduced integration selected?

3. Enter the stress-strain curve for copper as material number 1 using Multilinear Isotropic Hardening (MISO). Why is MISO selected?

4. Enter the coefficient of friction ($\mu = 0.3$), also using material 1.

Exercise

Steps to Follow:

5. *Without* using the Contact Wizard, create the target and contact elements as shown *using real constant set number 2.* Which side should be the target? Set the penalty stiffness scale factor (FKN) appropriately. Is the problem dominated by bending or bulk deformation?

6. Set the real constants PMAX and PMIN to move the rigid target surface into initial contact (PMAX = 0.1 and PMIN=0.01). Why is this necessary?

7. Create a pilot node at keypoint 23 (KMESH) to load the rigid surface.

8. Apply 20,000 lb/in in the -Y direction at the pilot node. (There are 3 nodes at this location. Only the pilot node is attached to the KP. Therefore, apply the force to the KP.)

Exercise

Steps to Follow:

9. Turn on nonlinear geometry (NLGEOM,ON).
10. Specify the number of substeps for automatic time stepping (20,500,10).
11. Specify the output controls (request every substep).
12. Run the solution, and review the output and the monitor files.

Exercise

Steps to Follow:

13. Postprocess the contact results, review contact pressures, and frictional stresses. Plot the von Mises stress and the accumulated equivalent plastic strain. Are the strain values beyond the supplied material data? Since we are using reduced integration elements be sure to compare the artificial energy and the strain energy.



Shear stresses under the applied load.

Exercise

Extra Credit

If you have time, restart the analysis and ramp the applied load back to zero. Review the permanent deformation and residual stresses in the copper plate.



von Mises stresses with the load removed

Exercise

Answers to Questions:

Since this is a large strain solution including both plasticity and contact, for maximum solution efficiency Plane182 with reduced integration is selected.

Since the problem is a large strain solution with proportional loading, multilinear isotropic hardening is selected.

The rigid surface is *always* the target surface.

This problem is dominated by bulk deformation, set FKN = 1.0.

Exercise

Answers to Questions:

Since the two bodies are initially unconnected, rigid body motion will be a concern. Setting PMAX and PMIN establishes the initial contact condition for the problem.

The strain values are beyond the data values supplied for the material. ANSYS assumes perfect plasticity beyond the last data point.

Solution to Exercise

```
THICK=0.25
RADIUS=0.75
CONTACT=0.3
LENGTH=1.0
/PREP7
ET,1,182,1,,2
ET,2,169
ET,3,171
R,2,0,0,1,0.1
RMORE,0.1,0.01
!*
!* Material
Properties
!*
MP,EX,1,16E6
MP,NUXY,1,0.33
MP,MU,1,0.3
TB,MISO,1,0,9
TBPT,,0.000625,100
00
TBPT,,0.0025,15000
TBPT,,0.005,21000
TBPT,,0.01,29000
TBPT,,0.015,32600
TBPT,,0.02,34700
TBPT,,0.04,36250
TBPT,,0.10,39000
TBPT,,0.20,40250
K,1
K,2,CONTACT
K,3,LENGTH
KGEN,2,1,3,1,,THICK,
,3
A,1,2,5,4
A,2,3,6,5
TYPE,1
MAT,1
LESIZE,4,,,10
LESIZE,2,,,10
LESIZE,6,,,10
LESIZE,1,,,12
LESIZE,3,,,12
LESIZE,5,,,12,5
LESIZE,7,,,12,1/5
AMESH,ALL
ARSYM,X,ALL
NUMMRG,ALL
ASEL,NONE
LSEL,NONE
WPAVE,,THICK+RAD
IUS
CSWPLA,11,1
K,20,RADIUS,180
K,21,RADIUS,-90
K,22,RADIUS,0
K,23,0,0
L,20,21
L,21,22
TYPE,2
REAL,2
LMESH,ALL
KMESH,23
KSEL,S,KP,,23
NSLK,S
*GET,N_LOAD,NODE,,NUM,
MAX
ALLS
CSYS,0
NSEL,S,LOC,Y,THICK
TYPE,3
ESURF
!*
!* Boundary Conditions
!*
NSEL,S,LOC,Y,0
D,ALL,UY
NSEL,S,LOC,X,0
NSEL,R,LOC,Y,0
D,ALL,UX
NSEL,ALL
FINISH
/SOLU
ANTYPE,STATIC
NLGEOM,ON
SOLC,ON
NSUBST,20,500,10
OUTRES,ALL,ALL
F,N_LOAD,FY,-
20000
SOLVE
F,N_LOAD,FY,0
SOLVE
FINI
/EXIT
```

NOTES

ANSYS, Inc.
Southpointe
275 Technology Drive
Canonsburg, PA 15317

ansysinfo@ansys.com
T 724.746.3304
F 724.514.9494
Toll Free USA and Canada:
1.800.WE.R.FEA.1
Toll Free Mexico:
95.800.9373321

Regional Offices
North America
len.zera@ansys.com
T 248.585.5020
F 248.585.5730
International
jim.tung@ansys.com
T 724.514.3086
F 724.514.3115
Europe
brian.butcher@ansys.com
T 44.118.9880229
F 44.118.9880925

http://www.ansys.com



001156